SPICE ANALYSIS OF LINEAR RESISTOR CIRCUITS WITH CONSTANT INPUTS

FALL 1995 **A.P. FELZER**

To analyze a linear resistor circuit like the following with SPICE



we must write a data file that tells SPICE what the circuit is and what analysis we want done

1. The **first** line of your data file **must** be a title such as

   CKT 1

   that **starts** with a letter and contains only letters and numbers. Always make sure that your first line is in fact your title and **not** a blank line

2. Next comes a description of the circuit. To accomplish this we must choose a reference node and then label the nodes and circuit elements as follows



   In doing this we have
   a. Arbitrarily numbered the nodes – except for the reference – which we **must** label 0
   b. Named the circuit elements in accordance with the following rules
      (i) Resistors start with the letter **R** and are followed by at most seven letters and numbers
      (ii) Voltage sources start with the letter **V** and are followed by at most seven letters and numbers
      (iii) Current sources start with the letter **I** and are followed by at most seven letters and numbers

   Once we have labeled our circuit we specify each circuit element with a line containing its name, the nodes it's connected to and its value. Adding this to the first line containing the title we have for our circuit

```
CKT1
Vin  1  0  DC  5
R1  1  2  1000
R2  2  3  2E3
RL  3  0  3K
```

3. And finally with the addition of the statement .END to signify the end of the data file we have a complete SPICE file as follows

```
CKT1
Vin  1  0  DC  5
R1  1  2  1000
R2  2  3  2E3
RL  3  0  3K
.END
```

The result of running this SPICE data file will be the circuit's node voltages and the currents thru the DC voltage sources

SOME IMPORTANT POINTS TO NOTE AND REMEMBER

1. We do not have to give a specific instruction for SPICE to do DC analysis – it always finds the circuit's node voltages due to the constant sources.

2. The reference directions of the sources are specified as follows –
   a. Voltage Sources – the first node listed is connected to the plus terminal
   b. Current Sources – the nodes are listed in the order the reference arrow is pointing

3. SPICE assumes voltages are in volts, currents are in amps and resistance is in ohms

4. There must be spaces between the entries as indicated and no spaces where not indicated

5. SPICE allows us to write powers of 10 in exponential form. We can, for example, write

   1000 as 1E3 (not 10E3)
   2,600,000 as 2.6E6

6. SPICE recognizes the following abbreviations –

SYMBOL	POWER OF 10
K	10^3
MEG	10^6
M	10^{-3}
U	10^{-6}

7. Separate lines of comments may be inserted anywheres between the title and .END. Such comments must always start with an asterisk * like the following comment

```
* SPICE sure beats hand calculations
```

Comments may also be appended to a line after a semicolon as follows

```
RL  1  2  1K     ; RL is the load resistor
```

8. To get SPICE to calculate and then print DC circuit element voltages and currents like those specified in our circuit as follows



we add the following line to our data file

```
.PRINT  DC  I(R2)  V(R2)
```

Note that the reference directions for I_{R2} and V_{R2} correspond to the order of the nodes which was first 2 and then 3 in the data entry for R_2

9. With the addition of the statement .TF V(3) Vin to the data file as follows

```
CKT1
Vin  1  0  DC  5
R1  1  2  1000
R2  2  3  2E3
RL  3  0  3K
.TF  V(3)  Vin
.END
```

SPICE will calculate

a. The transfer function $G = V(3)/V_{in}$
b. The equivalent resistance $R_{eq} = V_{in}/I_{in}$ as seen by the source V_{in}
c. And the Thevenin Equivalent resistance R_{TH} as seen by the output resistor R_L at node 3 – the equivalent resistance of



IMPORTANT THINGS FOR BEGINNERS TO REMEMBER

1. **Always** draw your circuit with nodes and circuit elements labeled before trying to write the data file
2. The first line of the data file must be your title – not a blank line
3. The title starts with a letter
4. Comments on separate lines have asterisks in front of them
5. .END has a period in front of it
6. All reference directions must be specified correctly
7. In SPICE R = 1E6 = 1MEG and not 1M

SPICE ANALYSIS OF LINEAR RESISTOR CIRCUITS WITH TIME-VARYING INPUTS

FALL 1995 **A.P. FELZER**

SINUSOIDAL INPUTS – To find the response of the following linear resistor circuit



to the sinusoidal input $v_{in}(t) = 5 \cos(2\pi 1000t + \pi/4) = 5 \sin(2\pi 1000t + 135)$ we must modify our data file from the previous example as follows

```
ResCkt
Vin  1  0  sin (0   5   1000   0   0   135)
R1  1  2  1000
R2  2  3  2E3
RL  3  0  3K
.TRAN  0.01ms  2ms
.PROBE
.END
```

1. The sinusoid $5 \sin(2\pi 1000t + 135)$ is specified by

   sin (0 5 1000 0 0 135)

   as follows –
   a. The first 0 means there is no DC offset. If $v_{in}(t)$ had been $v_{in}(t) = 2 + \cos(2\pi 1000t + \pi/4)$ then the 0 would have been replaced by a 2
   b. The 5 corresponds to the fact that the magnitude of the sinusoid is equal to 5
   c. The 1000 is the frequency of the sinusoid in **Hertz**. **All frequencies** in SPICE are in Hertz. It's very common for beginners to forget this fact – so beware
   d. The next 0 means that there is no time delay. The input $v_{in}(t)$ starts at time $t = 0$
   e. The final 0 means that $v_{in}(t)$ is a pure sinusoid. It is not a damped sinusoid – a sinusoid that is multiplied by a decaying exponential
   f. The 135 is the phase in **degrees**. **All phases** in SPICE are in degrees. It's very common for beginners to forget this fact – so beware

2. The TRAN statement

   .TRAN 0.01ms 2ms

   tells SPICE to analyze the circuit as a function of time
   a. The first number 0.01ms tells SPICE to find the circuit's response every 0.01 msec starting at time $t = 0$. Note that SPICE always starts its analysis at time zero. Note also that if this first number 0.01ms is not small enough, then our plots of the circuit's voltages and currents will not be smooth curves corresponding to the circuit's actual response
   b. The second number 2ms tells SPICE to do the analysis up to time $t = 2$ msec
3. The PROBE statement

.PROBE

tells SPICE to save the results of its calculations in a file Probe.DAT so that they can be used to plot the circuit's voltages and currents. To actually get a plot of the circuit's response

a. First select probe from the main menu
b. Then select Add_Trace from the Probe menu
c. And finally type in the names of the variable or variables you would like to plot and press return

PULSE TRAIN INPUTS – To find the response of our linear resistor circuit to a pulse train like



we replace the line Vin 1 0 sin (0 5 1000 0 0 135) in the data file by

Vin 1 0 PULSE (0 5 0 0.1us 0.1us 0.5ms 1ms)

1. The pulse train is specified by PULSE (V1 V2 td tr tf pw per) where
   a. V1 = 0 is the value of the pulse train at time t = 0 (just before the first pulse starts)
   b. V2 = 5 is the value of the pulse train when the pulse switches values
   c. td = 0 is the time delay. The time delay is the amount of time SPICE waits for the pulse train to start
   d. tr = 0.1us is the time it takes the pulse train to rise from its lower value of 0 volts to its higher value of 5 volts
   e. tf = 0.1us is the time it takes the pulse train to fall from its higher value of 5 volts to its lower value of 0 volts
   f. pw = 0.5ms is the pulse width of the pulses
   g. per = 1ms is the period of the pulse train

2. The .TRAN and .PROBE statements are the same as for the sinusoid

SPICE ANALYSIS OF CIRCUITS CONTAINING LINEAR CONTROLLED SOURCES AND OP AMPS

FALL 1995 A.P. FELZER

CONTROLLED SOURCES – For SPICE to analyze a circuit containing a controlled source such as the following



we must specifiy

(1) What kind of controlled source is being used
(2) Which voltage or current is doing the controlling
(3) Where the controlled source is located

Controlled sources are specified as follows

1. Voltage Controlled Voltage Source




```
E1  3  4   1  2   100
```

2. Voltage Controlled Current Source




```
G1  3  4   1  2   5
```

3. Current Controlled Voltage Source




```
VC1  1  2   DC   0
H1  3  4   VC1   10
```

4. Current Controlled Current Source




```
VC1  1  2   DC   0
F1  3  4   VC1   50
```

NOTE THAT

1. E is for voltage controlled voltage sources
   G is for voltage controlled current sources
   H is for current controlled voltage sources
   F is for current controlled current sources

2. The location of the controlled source is specified first and then the location of the controlling variable. Reference directions are very, very important in the specification of these locations. Be careful

3. Zero-valued independent voltage sources in general need to be inserted as shown above when our circuits have current controlled dependent sources

EXAMPLE – For SPICE to analyze the circuit above with a current controlled current source we must add a zero-valued independent voltage source as follows



The SPICE data file is then

```
Current Controlled Current Source
Vin   1  0   DC   5mv
R1   1  2   1K
VCS  2  3   DC   0
R2   3  0   1K
R3   2  4   5K
F1   4  0   VCS   100
RL   4  0   1K
.end
```

OP AMP CIRCUITS – One way to analyze a circuit containing an op amp like the following



is to simply replace the op amp by a simple model like the following



and then analyze the resulting circuit



just like any circuit containing a controlled source

Alternatively we can use one of the op amp macromodels that comes with PSPICE to analyze our circuit. Note that macromodels for op amps and other integrated circuits are equivalent circuits that exhibit the main characteristics of the actual circuit without containing as many circuit elements and therefore not requiring as long to analyze. For our op amp circuit



with the DC biasing supplies added we have the SPICE data file

```
Op Amp Circuit
Vin   1  0   sin (0  5  1000  0  0  0)
VCCpos   4  0   15
VCCneg   5  0   –15
R1   1  2   1K
R2   2  3   2K
XAMP   0  2  4  5  3    UA741
. LIB   EVAL.LIB
.TRAN   0.1ms   2ms
.PROBE
.END
```

where XAMP is PSPICE's macromodel for the 741 op amp. It should be clear by inspection what the numbers correspond to

SPICE ANALYSIS USING SUBCIRCUITS

FALL 1995 **A.P. FELZER**

Suppose the following circuit



is made from two copies of circuit N as given by



Then by using the method of **subcircuits** we only have to write the SPICE data code for N once. The procedure is as follows

1. First assign node numbers – except for 0 – to the nodes of N as follows



and then write a SPICE data file for subcircuit N in the following format

```
.SUBCKT  N  1  3  4
RA   1  2   1K
RB   2  4   2K
RC   3  4   3K
GA   3  4   2  4  0.2
.ENDS  N
```

2. Then include the data file for the subcircuit N in the SPICE data file for the whole circuit



as follows

```
Sub
.SUBCKT  N  1  3  4
RA   1  2   1K
RB   2  4   2K
RC   3  4   3K
GA   3  4   2  4  0.2
.ENDS  N
Vin   1  0   DC   10
R1   1  2   1K
X1   2  3  0   N
X2   3  4  0   N
R2   4  0   1K
.PROBE
.END
```

NOTE THAT –

1. X is for subcircuit

2. The name given to the subcircuit in this example is N

3. The subcircuit can have the same node numbers as the complete circuit except for 0 which only the reference of the complete circuit can have

4. The order of the nodes in the specification of X1 and X2 is very important. It specifies how the subcircuit is connected to the complete circuit. It tells SPICE what is connected to what

SPICE TRANSIENT ANALYSIS

FALL 1995 **A.P. FELZER**

The SPICE data file for finding the response of the following circuit



with initial conditions $v_C(0) = 2$ volts and $i_L(0) = -2$ ma as a function of time is as follows

```
Transient
Vin   1  0   DC  10
R1   1  2   300
L1   2  3   0.1   IC = –2ma
C1   3  0   0.1u   IC = 2
.TRAN   5us   1ms   UIC
.PROBE
.END
```

NOTE THAT –

1. Capacitors start with the letter C and inductors with the letter L

2. The line with the command .TRAN tells SPICE to obtain the circuit's time response every 5 μsec from $t = 0$ to $t = 1$ ms. UIC tells SPICE that the initial conditions for the capacitor and inductor are being specified by the user

3. The initial conditions IC for capacitors are the voltages across them at time $t = 0$ and for inductors the currents thru them at time $t = 0$. Default values for the initial conditions are 0. Be sure to note that the reference directions for the initial voltages and currents correspond to the order in which the nodes of L and C are specified

4. In general it can be difficult to estimate how long it will take a circuit's transient response to decay and therefore how small to make the time increment in order to get a good plot of the circuit's response. We want to make the increment small enough so we get an accurate graph but not so small that the analysis may take a longer time and more memory to save the results than necessary. You may just have to try your best guess for the increment and then try a smaller value to make sure that the graph of the response does not change

SPICE ANALYSIS OF CIRCUITS IN THE SINUSOIDAL STEADY STATE

FALL 1995 **A.P. FELZER**

The SPICE data file for finding the sinusoidal steady state response of the following circuit



as a function of frequency plotted on a log scale with 15 points/decade from f = 10 Hz to f = 100 KHz is as follows

```
Sinusoidal Steady State Response
Vin  1 0  AC 10
C  1 2  0.1u
L1  2 3  0.1
R1  3 0  300
.AC  DEC 15  10 1E5
.PROBE
.END
```

NOTE THAT –

1. AC tells SPICE to do sinusoidal steady state analysis. The 10 following the AC is the magnitude of the sinusoid

2. No initial conditions are specified in steady state analysis

3. The line with the command .AC tells SPICE to calculate the magnitude and phase of the circuit's sinusoidal steady state response at 15 points/decade from f = 10 Hz to f = 10^5 Hz. The statement .AC lin 500 10Hz 1E5 would have had SPICE calculate the frequency response on a linear scale starting at 10Hz and then increasing in steps of 500 Hz up to 100 KHz

4. Frequency analysis must always start at a frequency f > 0

5. When using probe to graph the frequency response of a circuit, use M for magnitude and P for phase. For example use VM(3) and VM(2, 3) to get plots of the magnitude of V(3) and V(2, 3) as a function of frequency, IP(R1) to get a plot of the phase of the current flowing thru resistor R1 and VM(3)/VM(1) to get a plot of the magnitude of the corresponding voltage transfer function

6. To find the sinusoidal steady state response at node 3 at a particular frequency – say at f = 100Hz simply replace the .AC statement above by

```
.AC lin 100 100
.PRINT V(3)
```

SPICE FOURIER ANALYSIS

FALL 1995 **A.P. FELZER**

The SPICE data file for finding the DC term and magnitudes and phases of the first nine harmonics of the Fourier Series expansions of the voltages and currents in the following circuit



is as follows where $v_{in}(t)$ is a pulse train of frequency f = 100KHz

```
Fourier Series Analysis
Vin  1  0   Pulse (1v  0v  0s  10ns  10ns  5ms  10ms)
R1   1  2   1K
R2   2  0   1K
.TRAN  50us  10ms
.FOUR   100Hz   V(1),  V(2)
.END
```

NOTE THAT –

1. A transient analysis must be done before SPICE can do its Fourier Series calculations because it uses the results of the transient analysis in its calculations
2. The statement instructing SPICE to do a Fourier Analysis is the statement .FOUR followed by the fundamental frequency in Hz and then the variables to be analyzed
3. SPICE will automatically print out the results of its Fourier calculations without any further commands
4. SPICE calculates the Fourier Coefficients with a numerical algorithm that depends on the number of points taken. A longer time will give better results but will require more memory space to execute